
FRBR-Strukturierung von MAB-Daten, oder: 

Wieviel MAB passt in FRBR? 

Die Expertengruppe MAB-Ausschuss (seit 2005: Expertengruppe Datenformate) 
hat sich im Verlauf des Jahres 2004 mit den FRBR und ihren Bezügen zum MAB-
Format befasst. Es wurde eine Tabelle FRBR => MAB erstellt (veröffentlicht 
unter http://www.ddb.de/professionell/pdf/frbr_mab.pdf), wichtige Ergebnisse 
wurden im Artikel "Maschinelles Austauschformat für Bibliotheken und die Func-
tional Requirements for Bibliographic Records : Oder: Wieviel FRBR verträgt 
MAB?" im "Bibliotheksdienst" 39 (2005), Heft 10 zusammengefasst. 

Ergänzend dazu wurde bei der Arbeitsstelle Datenformate Der Deutschen Biblio-
thek versucht, MAB-Daten zu "frbrisieren", d. h. einzelne MAB-Datensätze in die 
vier Entitäten der Gruppe 1 (work / expression / manifestation / item) zu diffe-
renzieren. Ziel war nicht, einen fertigen OPAC-Baustein für die Indexierung, 
Benutzerführung oder Präsentation zu erstellen. Ziel war vielmehr, anhand von 
konkreten, in MAB strukturierten Daten die Schichten sichtbar zu machen. 

Ausgewählt für diesen Zweck wurde BISMAS, das "Bibliographische Informati-
ons-System zur Maschinellen Ausgabe und Suche" des BIS Oldenburg 
(www.bismas.de). In BISMAS ist es mit relativ geringem Aufwand möglich, die 
Präsentation eines Satzes - basierend auf der intern vorliegenden Datensatzstruk-
tur, z.B. MAB - frei zu definieren. Die Gestaltung der Indices und der Ausgabe-
formate erfolgt in BISMAS mit Hilfe der Programmiersprache LM. Die Ergebnisse 
sollen hier anhand von Beispielen dargestellt werden. 

Es erschien zunächst zweckmäßig, zu den Entitäten der ersten Gruppe eine 
zusätzliche Schicht 0 "record" für den MAB-Datensatz zu definieren. In ihr wird 
ausgewertet, welche Angaben über den Datensatz selbst enthalten sind, also 
Satzstatus und Satztyp aus der Satzkennung, der Bearbeitungsstatus in 030 
Pos. 0 und die Erscheinungsform in 051 Pos. 0 bzw. 052 Pos. 0. 

Für die erste Ebene "work" sind die Primärverfasser in 100_ oder 200_ und der 
Einheitssachtitel (ohne die Sprachbezeichnung) in Feld 304, alternativ der Haupt-
sachtitel in Feld 331 relevant. Hinzu kommen die Originalsprache in Feld 038b 
und die Sacherschließungsdaten, hier die Notation in 700 und die RSWK-Anga-
ben in 9--. Einschränkend zu Schlagwörtern ist zu sagen, dass Formschlagwörter 
nur teilweise zur Ebene "work" gehören: Wenn sie die physikalische Form betref-
fen, sind sie der Ebene der "manifestation" zuzuordnen. 

Die zweite Ebene "expression" ist in den Daten nicht sehr stark vertreten. Es 
gehören hierzu der Hauptsachtitel in 331, wenn ein EST in 304 vorhanden ist, 
außerdem die Sprache des Textes in 037b und in 304 innerhalb der Spitzklam-
mern, sowie Übersetzer und Bearbeiter in 1XX. Zusätzlich können bei Daten 
nach RAK-Musik alle Interpreten, also alle Angaben in 1XXe und 2XXe, der 
Ebene der "expression" zugeordnet werden. 

Breiten Raum nimmt die dritte Ebene "manifestation" ein. Hier sind die Angaben 
zum Datenträger in Feld 050, die Formschlagwörter (mit der genannten Ein-
schränkung), der Ländercode in 036a, das gesamte Segment 400 - 437 "Veröf-
fentlichungsvermerk, Umfang, Beigaben", die ISBN in 544 und die EAN in 553 
zu nennen. 

Wenn das Feld 544 für die Signatur verwendet wird, ist auch die vierte Ebene 
"item" vertreten, die sonst in MAB-LOKAL abgebildet wird. 

Folgende Einschränkungen werden deutlich: FRBR kann nur bedingt als echtes 
Präsentationsformat dienen. Einem Benutzer sind die verschiedenen Ebenen nicht 
verständlich. Zudem fehlen alle Angaben, die keiner der Ebenen zuzuordnen sind, 
also z.B. die Verfasserangabe in Feld 359, die meisten Fußnoten, oder auch die 
eher inhaltlich ausgerichteten Codierungen in den Feldern 051 und 052. 

Wie die Statistik in Anhang 1 zeigt, ist in deutschen MAB-Daten der Einheits-
sachtitel in Feld 304 zu selten vertreten, um als durchgehendes Merkmal der 
Ebene "work" zu dienen. Ähnliches gilt bei der eher seltenen Angabe von Über-
setzern für die Ebene der "expression" und bei der fehlenden Normierung von 
Ausgabebezeichnungen und Verlagsnamen für die Ebene der "manifestation". 

Ein großes Problem stellt der Umstand dar, dass in MAB mehrbändige begrenzte 
Werke mit Bandaufführungen als bibliografische Hierarchien dargestellt werden. 
Wie das Beispiel 5 zeigt, ist weder ein h-Satz noch ein u-Satz (noch gar ein y-
Satz für eine unselbständige Abteilung!) für sich genommen in FRBR-Sicht aus-
sagekräftig genug. Für eine vollständigere Darstellung wäre es nötig, die Anga-
ben aus den miteinander verknüpften Sätzen zusammenzuführen. 

In Absprache mit dem BIS Oldenburg kann zu Testzwecken das Programm 
BISMAS mit den benötigten Parameter- und Steuerdateien sowie einer Anleitung 
zusammen mit Testdaten Der Deutschen Bibliothek (insgesamt ca. 2,7 MB) bei 
der Arbeitsstelle Datenformate, heuvelmann@dbf.ddb.de, angefordert werden. 

Hinweis: 

Einen breiteren Ansatz als das hier vorgestellte Verfahren haben Knut Hegna und 
Eeva Murtomaa mit ihrer Studie "Data mining MARC to find: FRBR?" 
(http://folk.uio.no/knuthe/dok/frbr/) verfolgt. Herr Hegna hat dankenswerterweise 
das "MarcStudy Tool" dahingehend modifiziert, dass es MAB-Daten im Disket-
tendienst-Format mit der Erweiterung ".mab" einliest, auswertet und nach FRBR 
strukturiert navigierbar macht. Es handelt sich um eine Java-Application, die 
online unter "http://folk.uio.no/knuthe/progdist/" erreichbar ist. Die Anleitung 
zum MarcStudy Tool unter "http://folk.uio.no/knuthe/dok/frbr/userguide.pdf" gilt 
analog. Auch hierzu können von der Arbeitsstelle Datenformate, 
heuvelmann@dbf.ddb.de, Testdaten zur Verfügung gestellt werden. 

Für MARC-21-Daten bietet die Library of Congress ein Visualierungstool an 
(siehe: http://www.loc.gov/marc/marc-functional-analysis/tool.html). 

Beispiel 1: einfache Monografie 

1a: in MAB-Struktur 

### 00839nM2.01200024      h 
001 970846126 
002a20040408 
004 20041207 
030 g|1iaz|z|||17 
036aXA-DE-BY 
037bger 
050 a|b||||||||||| 
051 s|||z||| 
070a292 
100 Kafka, Franz 
102a118559230 
331 Amerika 
335 Roman 
359 Franz Kafka 
410 München 
412 Süddt. Zeitung 
425 2004 
425a2004 
433 279 
435 21 cm 
451 Bibliothek / Süddeutsche Zeitung ; [36] 
453m970738234 
454bSüddeutsche Zeitung GmbH <München>: Bibliothek 
455 [36] 
456 236 
501 Lizenz 
540aISBN 3-937793-35-6 Pp. : EUR 4.90 (DE), EUR 5.10 (AT), sfr 
    8.90 
544aDBF#2004 A 67787 
553a9783937793351 
568 04,N19,0641 
574 04,A52,1739 
700 |830 
700 |B 

1b: in FRBR-Struktur, basierend auf MAB 

-------------------------------------------------------------- 
0 MAB -- Satzstatus n -- Satztyp h -- vervollständigt CIP ---- 
    Erscheinungsform: Einbändiges Werk, Stücktitel ----------- 
-------------------------------------------------------------- 
1 Work ------------------------------------------------------- 
           1. Person: Kafka, Franz 
      Hauptsachtitel: Amerika 
            Notation: 830 
            Notation: B 
2 Expression ------------------------------------------------- 
        Sprachencode: ger 
3 Manifestation ---------------------------------------------- 
         Datenträger: Druckschrift; säurefreies Papier 
          Ländercode: XA-DE-BY 
   Orte des 1. Verl.: München 
       Erster Verlag: Süddt. Zeitung 
    Erscheinungsjahr: 2004 
    Erscheinungsjahr: 2004 
              Umfang: 279 
              Format: 21 cm 
                ISBN: 3-937793-35-6 
       Artikelnummer: 9783937793351 
4 Item ------------------------------------------------------- 
        Signatur DDB: DBF#2004 A 67787 
-------------------------------------------------------------- 

Beispiel 2: Übersetzung 

2a: in MAB-Struktur 

### 00001nM2.01200024      h 
001 00155925 
002a20030411 
003 20031105 
026f10787060 
030 a|5||r|zc|||| 
036aCZ 
037acs 
050 a|a||||||||||| 
051 sd|||||| 
100 Kafka, Franz 
104bČermák, Josef ¬[Übersetzer]¬ 
108bRóna, Jaroslav ¬[Illustrator]¬ 
304 Amerika <tschech.> 
331 Nezvěstný 
341 Amerika 
359 Franz Kafka. [Transl. Josef Čermák. Ill. Jaroslav Róna] 
403 1. vyd. 
410 Praha 
412 Nakl. Labyrint 
425 2003 
433 293 S. 
434 Ill. 
451 Edice Granada ; 6 
453 00174294 
455 6 
540 ISBN 80-85935-35-X 

2b: in FRBR-Struktur, basierend auf MAB 

-------------------------------------------------------------- 
0 MAB -- Satzstatus n -- Satztyp h -- Autopsie --------------- 
    Erscheinungsform: Einbändiges Werk, Stücktitel ----------- 
-------------------------------------------------------------- 
1 Work ------------------------------------------------------- 
           1. Person: Kafka, Franz 
   Einheitssachtitel: Amerika 
2 Expression ------------------------------------------------- 
      Hauptsachtitel: Nezvěstný 
   Einheitssachtitel: tschech. 
           2. Person: Čermák, Josef ¬[Übersetzer]¬ 
3 Manifestation ---------------------------------------------- 
         Datenträger: Druckschrift 
          Ländercode: CZ 
   Ausg. Vorlageform: 1. vyd. 
   Orte des 1. Verl.: Praha 
       Erster Verlag: Nakl. Labyrint 
    Erscheinungsjahr: 2003 
              Umfang: 293 S. 
      Illustrationen: Ill. 
                ISBN: 80-85935-35-X 
4 Item ------------------------------------------------------- 
-------------------------------------------------------------- 

Beispiel 3: Tonträger 

3a: in MAB-Struktur 

### 00001nM2.01000024 h 
001 00004510 
002a19990224 
003 19990408 
026f10689540 
030 k|5||r|zc|||| 
036aXA-DE 
037ade 
050 ||a||yy||||||| 
051 m||||||| 
100 Kafka, Franz 
104aBrod, Max 
108bCremer, Ludwig ¬[Bearbeiter]¬ 
112bZwetkoff, Peter ¬[Komponist]¬ 
116bTeege, Joachim ¬[Interpret]¬ 
120bBalthoff, Alfred ¬[Interpret]¬ 
124bRomberg, Rudolph ¬[Interpret]¬ 
331 Amerika 
359 von Franz Kafka und Max Brod. Funkbearb. und Regie: Ludwig  
    Cremer. Komp.: Peter Zwetkoff. Darst.: Joachim Teege,  
    Alfred Balthoff, Rudolph Romberg 
410 [Baden-Baden] 
412 Südwestfunk 
425 1957 
433 1 Tonkassette (DAT, 88 Min.) 
434 mono 

3b: in FRBR-Struktur, basierend auf MAB 

0 MAB -- Satzstatus n -- Satztyp h -- Autopsie --------------- 
    Erscheinungsform: Einbändiges Werk, nicht Stücktitel ----- 
-------------------------------------------------------------- 
1 Work ------------------------------------------------------- 
           1. Person: Kafka, Franz 
      Hauptsachtitel: Amerika 
2 Expression ------------------------------------------------- 
           3. Person: Cremer, Ludwig ¬[Bearbeiter]¬ 
3 Manifestation ---------------------------------------------- 
         Datenträger: Audiovisuelles Material 
          Ländercode: XA-DE 
   Orte des 1. Verl.: [Baden-Baden] 
       Erster Verlag: Südwestfunk 
    Erscheinungsjahr: 1957 
              Umfang: 1 Tonkassette (DAT, 88 Min.) 
      Illustrationen: mono 
4 Item ------------------------------------------------------- 
-------------------------------------------------------------- 

Beispiel 4: Monografie mit Übersetzer und RSWK-Feldern 

4a: in MAB-Struktur 

### 02626nM2.01200024      h 
001 972782109 
002a20041119 
004 20041207 
030 a|1iar|z|||17 
036aXA-FR 
037bfre 
038bger 
050 a|b||||||||||| 
051 m|||z||| 
070a292 
076 u| 
100 Mann, Thomas 
102a118577166 
104bServicen, Louise ¬[Übers.]¬ 
106a114199248 
304 Betrachtungen eines Unpolitischen <franz.> 
331 Considérations d'un apolitique 
359 Thomas Mann. Trad. de l'allemand par Louise Servicen et  
    Jeanne Naujac. Introd. de Jacques Brenner 
410 Paris 
412 Grasset 
425 2002 
425a2002 
433 487 S. 
435 23 cm 
540aISBN 2-246-01922-2 kart. : EUR 22.90 
544aDBL#F 2003 A 3881 
574 04,A52,0361 
700 |320 
700 |830 
902p  4037361-7           Mann, Thomas 
902s114115590-7           Politisches Denken 
902f11|Quelle 

4b: in FRBR-Struktur, basierend auf MAB 

-------------------------------------------------------------- 
0 MAB -- Satzstatus n -- Satztyp h -- Autopsie --------------- 
    Erscheinungsform: Einbändiges Werk, nicht Stücktitel ----- 
-------------------------------------------------------------- 
1 Work ------------------------------------------------------- 
           1. Person: Mann, Thomas 
   Einheitssachtitel: Betrachtungen eines Unpolitischen 
    Herkunftssprache: ger 
            Notation: 320 
            Notation: 830 
   Kettenglied 1. SW: Mann, Thomas   
   Kettenglied 1. SW: Politisches Denken   
   Kettenglied 1. SW: Quelle 
2 Expression ------------------------------------------------- 
      Hauptsachtitel: Considérations d'un apolitique 
        Sprachencode: fre 
   Einheitssachtitel: franz. 
           2. Person: Servicen, Louise ¬[Übers.]¬ 
3 Manifestation ---------------------------------------------- 
         Datenträger: Druckschrift; säurefreies Papier 
      Formschlagwort: Quelle 
          Ländercode: XA-FR 
   Orte des 1. Verl.: Paris 
       Erster Verlag: Grasset 
    Erscheinungsjahr: 2002 
    Erscheinungsjahr: 2002 
              Umfang: 487 S. 
              Format: 23 cm 
                ISBN: 2-246-01922-2 
4 Item ------------------------------------------------------- 
        Signatur DDB: DBL#F 2003 A 3881 
-------------------------------------------------------------- 

Beispiel 5: Mehrbändiges begrenztes Werk aus h-Satz und u-Satz 

5a: in MAB-Struktur 

h-Satz 

### 02560nM2.01200024      h 
001 972768815 
002a20041118 
004 20041207 
030 a|1iar|a|||17 
036aXA-YUCS 
037bscc 
038bger 
050 a|a||||||||||| 
051 n|||z||| 
070a292 
076 u| 
100 Spengler, Oswald 
102a118616110 
304 ¬Der¬ Untergang des Abendlandes <serb.> 
331 Propast Zapada 
359 Osvald Spengler. [Preveo Vladimir Vujic] 
410 Beograd 
412 Utopija 
435 22 cm 
501 In kyrill. Schr., serb. 
540aISBN 86-84175-12-3 
700 |300 
700 |100 
902s  4120625-3           Kulturmorphologie 

u-Satz 

### 02561nM2.01200024      u 
001 972768890 
002a20041118 
004 20041207 
010 972768815 
030 a|1ia||a||||| 
037bscc 
038bger 
050 a|b||||||||||| 
070a292 
076 u| 
089 Knj. 1. 
090 11 
400 11 
403 1. izd. 
425 2003 
425a2003 
433 582 S. 
540aISBN 86-84175-11-5 Pp. 
544aDBL#F 2003 A 3996 
574 04,A52,0332 

5b: in FRBR-Struktur, basierend auf MAB 

h-Satz 

-------------------------------------------------------------- 
0 MAB -- Satzstatus n -- Satztyp h -- Autopsie --------------- 
    Erscheinungsform: Mehrbändiges begr. Werk, nicht Stückt. - 
-------------------------------------------------------------- 
1 Work ------------------------------------------------------- 
           1. Person: Spengler, Oswald 
   Einheitssachtitel: ¬Der¬ Untergang des Abendlandes 
    Herkunftssprache: ger 
            Notation: 300 
            Notation: 100 
   Kettenglied 1. SW: Kulturmorphologie   
2 Expression ------------------------------------------------- 
      Hauptsachtitel: Propast Zapada 
        Sprachencode: scc 
   Einheitssachtitel: serb. 
3 Manifestation ---------------------------------------------- 
         Datenträger: Druckschrift 
          Ländercode: XA-YUCS 
   Orte des 1. Verl.: Beograd 
       Erster Verlag: Utopija 
              Format: 22 cm 
                ISBN: 86-84175-12-3 
4 Item ------------------------------------------------------- 
-------------------------------------------------------------- 

 

u-Satz 

-------------------------------------------------------------- 
0 MAB -- Satzstatus n -- Satztyp u -- Autopsie --------------- 
-------------------------------------------------------------- 
1 Work ------------------------------------------------------- 
    Herkunftssprache: ger 
2 Expression ------------------------------------------------- 
        Sprachencode: scc 
3 Manifestation ---------------------------------------------- 
         Datenträger: Druckschrift; säurefreies Papier 
   Ausg. Sortierform: 11 
   Ausg. Vorlageform: 1. izd. 
    Erscheinungsjahr: 2003 
    Erscheinungsjahr: 2003 
              Umfang: 582 S. 
                ISBN: 86-84175-11-5 
4 Item ------------------------------------------------------- 
        Signatur DDB: DBL#F 2003 A 3996 
-------------------------------------------------------------- 

Anhang 1: Feldstatistiken mit den 60 häufigsten Felder in DNB, DMA und ZDB * 
DNB-CD-ROM  DMA-CD-ROM ZDB-Gesamtabzug 

Feldnummer Anzahl Feldnummer Anzahl Feldnummer Anzahl 
001 713900 001 604185 001 1119885
002 713900 002 604185 002 1119885
004 713900 004 604185 003 404771
010 94761 010 43338 004 1119885
038 71512 029 432401 025 1119885
036 611270 030 604185 030 1119885
037 636387 036 405899 036 1094604
030 713900 050 604185 037 734316
052 62661 051 579356 050 1119885
051 570331 070 604185 052 1119885
050 713900 089 42033 057 25034
076 63346 090 43338 070 1119885
070 713900 100 460463 200 618230
089 83265 102 460463 202 618230
090 94762 104 143593 204 64951
108 28428 106 143593 206 64951
106 90571 108 63023 310 132724
104 105597 110 63023 331 1119885
102 472813 112 40594 333 175901
100 512814 114 40594 334 32888
110 22480 116 26196 335 404572
202 81279 118 26196 341 65874
200 83155 120 17304 342 10747
304 47465 122 17304 345 10668
334 44249 200 181717 359 402099
335 351936 202 181717 360 132281
331 674699 204 45202 370 329091
359 567026 206 45202 376 38316
370 34687 208 19993 405 1119885
405 33979 210 19993 410 1119885
403 213330 304 224244 412 582013
400 216163 331 435765 418 51292
412 534878 335 136878 425 1117447
410 534879 341 17919 451 31279
425 662388 359 165498 503 34445
434 374004 361 62500 505 72342
435 567456 370 232722 507 130016
433 585273 403 94794 509 44856
453 164326 410 414291 510 48676
455 171423 412 414291 516 69502
454 172198 415 51548 523 117869
456 174736 417 51548 524 20118
451 264687 425 582318 527 71017
501 183235 433 420949 529 25874
519 126405 434 63562 530 37959
536 47417 435 186684 531 253691
542 24982 437 224721 532 33983
540 525082 451 58455 533 255283
544 572684 501 243840 534 102179
553 134525 510 15232 537 453378
568 170062 511 66387 542 210296
574 648227 512 23410 580 14017
653 22442 517 225781 610 23962
655 23948 518 89548 611 12169
652 37221 541 19096 613 12093
700 618524 544 427370 652 32637
908 48281 551 416466 655 25872
907 73013 574 440738 673 66040
903 205752 599 154948 675 18181
902 332127 700 406311 700 1119885

 
* Die Auswertung erfolgte auf der Grundlage des folgenden Datenmaterials: 
DNB-CD-ROM 2004/1, DMA-CD-ROM 2004/1, ZDB-Gesamtabzug Februar 2004 

Anhang 2: BISMAS-Ausgabesteuerung "MAB FRBR" 

MAB FRBR 
 
set lineindent 0; 
 
write "0 MAB -- "; 
 
for "000 " do { 
   skip 5; 
   write "Satzstatus "; 
   write 1; 
   write " -- "; 
   skip - 6; 
   if      cat$[23;1] = "h" then { 
      write "Satztyp h"; 
   } 
   else if cat$[23;1] = "y" then { 
      write "Satztyp y"; 
   } 
   else if cat$[23;1] = "u" then { 
      write "Satztyp u"; 
   } 
} 
 
write " -- "; 
 
for "030 " do { 
        if cat$[0;1] = "a" then write "Autopsie --------------"; 
   else if cat$[0;1] = "b" then write "teilweise Autopsie ----"; 
   else if cat$[0;1] = "c" then write "Übernahme aus Nationalbibliographie"; 
   else if cat$[0;1] = "d" then write "Übernahme aus anderen Quellen"; 
   else if cat$[0;1] = "e" then write "konvertierte Altdaten -"; 
   else if cat$[0;1] = "f" then write "CIP-Aufnahme ----------"; 
   else if cat$[0;1] = "g" then write "vervollständigt CIP ---"; 
   else if cat$[0;1] = "h" then write "ohne Autopsie ---------"; 
   else if cat$[0;1] = "u" then write "maschinelle Umsetzung -"; 
   else if cat$[0;1] = "z" then write "-----------------------"; 
   else if cat$[0;1] = "|" then write ""; 
} 
 
writeline "-"; 
 
for "000 " do { 
   if cat$[23;1] = "h" then { 
      write "    Erscheinungsform: "; 
 
      for "051" do { 
              if cat$[0;1] = "a" then write "Unselbständig erschienenes Werk --------"; 
         else if cat$[0;1] = "f" then write "Fortsetzung ----------------------------"; 
         else if cat$[0;1] = "h" then write "Finite Integrating Resource ------------"; 
         else if cat$[0;1] = "m" then write "Einbändiges Werk, nicht Stücktitel -----"; 
         else if cat$[0;1] = "n" then write "Mehrbändiges begr. Werk, nicht Stückt. -"; 
         else if cat$[0;1] = "s" then write "Einbändiges Werk, Stücktitel -----------"; 
         else if cat$[0;1] = "t" then write "Mehrbändiges begr. Werk, Stücktitel ----"; 
         else if cat$[0;1] = "|" then write "----------------------------------------"; 
      } 
 
      for "052" do { 
              if cat$[0;1] = "a" then write "Unselbständig erschienenes Werk --------"; 
         else if cat$[0;1] = "f" then write "Fortsetzung ----------------------------"; 
         else if cat$[0;1] = "i" then write "Continuing Integrating Resource --------"; 
         else if cat$[0;1] = "j" then write "Zeitschriftenartige Reihe --------------"; 
         else if cat$[0;1] = "p" then write "Zeitschrift ----------------------------"; 
         else if cat$[0;1] = "r" then write "Schriftenreihe (Serie) -----------------"; 
         else if cat$[0;1] = "z" then write "Zeitung --------------------------------"; 
         else if cat$[0;1] = "|" then write "----------------------------------------"; 
      } 
 
      writeline ""; 
 
   } 
} 
 
writeline "--------------------------------------------------------------"; 

writeline "1 Work -------------------------------------------------------"; 
 
set lineindent 3; 
 
for "100 " do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
for "200 " do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
if exist("304.") then { 
   for "304." do { 
      write catprompt$; 
      if catstr(" <")<>-1 then { 
         write < " <"; 
      } 
      else write; 
      writeline ""; 
   } 
} 
else { 
   for "331." do { 
      write catprompt$; 
      write; 
      writeline ""; 
   } 
} 
 
for "038b" do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
for "700 " do { 
   write catprompt$; 
   skip 1; 
   write; 
   writeline ""; 
} 
 
for "902.", "907.",  
    "912.", "917.", 
    "922.", "927.", 
    "932.", "937.", 
    "942.", "947." do { 
   write catprompt$; 
   set s$1 cat$[2;1]; 
   if      s$1="|" then skip 3; 
   else if s$1=" " then skip 22; 
   else                 skip 20; 
   write; 
   writeline ""; 
} 
 
set lineindent 0; 
 
writeline "2 Expression -------------------------------------------------"; 
 
set lineindent 3; 
 
if exist("304.") then { 
   for "331." do { 
      write catprompt$; 
      write; 
      writeline ""; 
   } 
} 
 
for "037b" do { 
   write catprompt$; 
   write; 
   writeline ""; 

} 
 
for "304." do { 
   if catstr(" <")<>-1 then { 
      skip >" <"; 
      write catprompt$; 
      write <">"; 
      writeline ""; 
   } 
} 
 
for "1." do { 
   if catstr("[Übers")<>-1 then { 
      write catprompt$; 
      write; 
      writeline ""; 
   } 
   if catstr("[Bearb")<>-1 then { 
      write catprompt$; 
      write; 
      writeline ""; 
   } 
} 
 
for "100e", "104e", "108e", "112e", "116e", 
    "120e", "124e", "128e", "132e", "136e", 
    "140e", "144e", "148e", "152e", "156e", 
    "160e", "164e", "168e", "172e", "176e", 
    "180e", "184e", "188e", "192e", "196e" do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
for "200e", "204e", "208e", "212e", "216e", 
    "220e", "224e", "228e", "232e", "236e", 
    "240e", "244e", "248e", "252e", "256e", 
    "260e", "264e", "268e", "272e", "276e", 
    "280e", "284e", "288e", "292e", "296e" do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
set lineindent 0; 
 
writeline "3 Manifestation ----------------------------------------------"; 
 
set lineindent 3; 
 
for "050 " do { 
   write "      Datenträger: "; 
   set s$1 cat$[0;1]; 
   if      s$1="a" then write "Druckschrift"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[1;1]; 
   if      s$1="a" then write "Handschrift"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[2;1]; 
   if      s$1="a" then write ""; 
   else if s$1="b" then write "; säurefreies Papier"; 
   else if s$1="c" then write "; kein säurefreies Papier"; 
   else if s$1="d" then write "; entsäuertes Papier"; 
   else if s$1="e" then write "; Pergament"; 
   else if s$1="z" then write "; sonstiges Material"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[3;1]; 
   if      s$1="a" then write "Mikroform"; 
   else if s$1="b" then write "Mikroform-Master"; 
   else if s$1="c" then write "Sekundärform"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[4;1]; 
   if      s$1="a" then write "Blindenschriftträger"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[5;2]; 
   if      s$1="aa" then write "Audio-CD (CD-DA)"; 
   else if s$1="ad" then write "Compact-Cassette"; 
   else if s$1="aj" then write "Schallplatte"; 
   else if s$1="ca" then write "Videobandkassette"; 
   else if s$1="cd" then write "Bildplatte (Videodisc)"; 
   else if s$1="yy" then write "Audiovisuelles Material"; 

   else if s$1="||" then write ""; 
   set s$1 cat$[7;1]; 
   if      s$1="a" then write "Medienkombination"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[8;1]; 
   if      s$1="a" then write "Computerdatei"; 
   else if s$1="b" then write "Diskette"; 
   else if s$1="c" then write "Magnetbandkassette"; 
   else if s$1="d" then write "Optische Speicherplatte"; 
   else if s$1="e" then write "Einsteckmodul"; 
   else if s$1="f" then write "Magnetband"; 
   else if s$1="g" then write "Computerdatei im Fernzugriff"; 
   else if s$1="z" then write "Sonstige Computerdatei"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[9;1]; 
   if      s$1="a" then write "Spiel"; 
   else if s$1="|" then write ""; 
   set s$1 cat$[10;1]; 
   if      s$1="a" then write "; Landkarte"; 
   else if s$1="|" then write ""; 
 
   writeline ""; 
 
} 
 
for "902f", "907f",  
    "912f", "917f", 
    "922f", "927f", 
    "932f", "937f", 
    "942f", "947f" do { 
   write "   Formschlagwort: "; 
   set s$1 cat$[2;1]; 
   if      s$1="|" then skip 3; 
   else if s$1=" " then skip 22; 
   else                 skip 20; 
   write; 
   writeline ""; 
} 
 
for "036a" do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
for "40.", "41.", "42.", "43.", "44." do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
for "540." do { 
   if catstr("ISBN ")<>-1 then { 
      skip >"ISBN "; 
      write catprompt$; 
      write <" "; 
      writeline ""; 
   } 
} 
 
for "553." do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
set lineindent 0; 
 
writeline "4 Item -------------------------------------------------------"; 
 
set lineindent 3; 
 
for "544." do { 
   write catprompt$; 
   write; 
   writeline ""; 
} 
 
set lineindent 0; 
 
writeline "--------------------------------------------------------------"; 


